

Technika Cyfrowa - Ćwiczenie 4

Mateusz Furga

Maj 2022

1 Zadanie

Celem zadania jest zbudowanie prostego kalkulatora pozwalającego na dodanie i pomnożenie dwóch 4-bitowych liczb opartego na dowolnym układzie FPGA.

2 Wprowadzenie do układów FPGA

FPGA (field-programmable gate array) to programowalne układy logiczne składające się z tysięcy konfiguracyjnych bloków logicznych (configurable logic blocks), które są odpowiednio ze sobą połączone aby otrzymać pożądaną układ. Sposób tych połączeń definiujemy w językach opisu sprzętu (HDL) takich jak: VHDL i Verilog. Stanowią one warstwę abstrakcji pomiędzy sprzętem oraz znacząco przyspieszają i ułatwiają projektowanie układu.

Najważniejsze zalety układów FPGA:

1. Pozwalają na budowę przeróżnych układów elektronicznych np. własnego procesora, układu kryptograficznego czy systemu czasu rzeczywistego.
2. Są o rzędy wielkości szybsze od mikrokontrolerów czy procesorów, ponieważ wykonują obliczenia równolegle oraz nie dekodują instrukcji kodu maszynowego. Sam "program" jest zapisany w postaci odpowiednich połączeń bramek logicznych.
3. Są bardzo elastyczne, ponieważ pozwalają na łatwą zmianę wewnętrznej konfiguracji układu FPGA, dzięki czemu wykonanie zmian w produkcie nie wymaga modyfikacji innych elementów oraz zmiany schematu.
4. Posiadają dużą liczbę portów I/O dlatego nowe układy są produkowane w technologii BGA, która pozwala ograniczyć ilość zajmowanego miejsca przez układ scalony.


Układy FPGA są wykorzystywane m.in. w:

1. W przetwornikach cyfrowo-analogowych oraz przetwarzaniu sygnałów.
2. Wojsku oraz lotnictwie.
3. Projektowaniu układów ASIC. Gotowe układy pracujące na FPGA przekształca się na dedykowane wykonując tak zwane twarde kopie.
4. Astronomii i kosmonautyce np. układy FPGA były wykorzystane w misji na Marsa w łazikach Spirit i Opportunity.

3 Realizacja zadania

3.1 Wykorzystany układ FPGA

Zadanie zrealizowałem na płytce Basys 2, który bazuje na FPGA firmy Xilinx Spartan 3E XC3S100E.



Rysunek 1: Płytki Basys 2.


W skład płytki Basys 2 wchodzi następujące elementy:

1. Xilinx Spartan 3E XC3S100E FPGA
2. Atmel AT90USB2 USB2 - dostarczający zasilanie oraz interfejs programistyczny
3. Xilinx Platform Flash ROM - pamięć Flash, w której znajduje się konfiguracja FPGA wgrawana przy starcie działania
4. 8 LEDs, 4 wyświetlacze siedmiosegmentowe, 4 przyciski (push-button), 8 przełączników (switchy)
5. Port PS/2 oraz 8 bitowy port VGA
6. Regulowany zegar o częstotliwości taktowania 25Mhz, 50Mhz i 100Mhz

Sam Xilinx Spartan 3E to niedrogi FPGA posiadający 240 układów CLB oraz 108 portów I/O.

3.2 Opis rozwiązania

Aby wprowadzić 2 liczby do kalkulatora korzystamy z 8 switchy, po 4 dla każdej z liczb co daje zakres 0 - 15 możliwych wartości. Wprowadzone liczby wyświetlają się w postaci dziesiętnej na 4 wyświetlaczach siedmiosegmentowych po 2 dla każdej z liczb.



Rysunek 2: Wprowadzenie 2 liczb: 11 i 6 oraz jednoczesne wyświetlenie ich na 4 wyświetlaczach siedmiosegmentowych.

Wprowadzone liczby l i r są przechowywane w 11-bitowym rejestrze bin w postaci: $100l + r$. Następnie przy pomocy modułu $bin2bcd$ liczba bin jest konwertowana na 16 bitową w reprezentacji BCD, która potem jest wyświetlana przez moduł $seg7$ na 4 wyświetlaczach siedmiosegmentowych.

W celu dodania dwóch liczb do siebie należy przytrzymać prawy przycisk (G12), co spowoduje że do rejestru bin powęduje suma: $l + r$. Natomiast aby pomnożyć 2 liczby należy przytrzymać lewy przycisk (C11), wtedy w rejestrze bin znajdzie się ich iloczyn: $l \cdot r$. Każda zmiana rejestru bin jest od razu widoczna na wyświetlaczach siedmiosegmentowych.

Moduł $bin2bcd$ konwertuje 11-bitową liczbę w postaci binarnej na 16-bitową w postaci BCD wykorzystując algorytm *double dabble*.

Moduł $seg7$ jest odpowiedzialny za kontrolę nad wyświetlaczem. Każdy z 4 wyświetlaczy jest podłączony do wspólnych 7 katod, które stanowią input do układu. Skutkiem tego jest powielanie tej samej cyfry na każdym z nich. Aby wyświetlić różne cyfry, moduł $seg7$ przełącza cyklicznie każdy z 4 wyświetlaczy, tak aby w danym momencie działał tylko jeden i wyświetlał odpowiednią cyfrę. Częstotliwość przełączania wyświetlaczy jest większa niż reakcja ludzkiego oka wynikiem czego na wyświetlaczu widzimy jednocześnie 4 cyfry.

3.3 Implementacja

Układ zaimplementowałem w języku Verilog. W celu wygenerowania pliku .bit użyłem programu Xilinx ISE 14.7 oraz dltgcfg firmy Digilent do zaprogramowania pamięci Flash.

```
1 module bin2bcd
2   #(parameter BUS_WIDTH = 11)(
3     input wire [BUS_WIDTH:0] bin,
4     output reg [15:0] bcd
5   );
6
7   integer i;
8
9   always @ (bin) begin
10     bcd = 0;
11     for (i = 0; i <= BUS_WIDTH; i = i + 1) begin
12       if (bcd[3:0] >= 5) bcd[3:0] = bcd[3:0] + 3;
13       if (bcd[7:4] >= 5) bcd[7:4] = bcd[7:4] + 3;
14       if (bcd[11:8] >= 5) bcd[11:8] = bcd[11:8] + 3;
15       if (bcd[15:12] >= 5) bcd[15:12] = bcd[15:12] + 3;
16       bcd = { bcd[14:0], bin[BUS_WIDTH - i] };
17     end
18   end
19 endmodule
```

Listing 1: Kod modułu bin2bcd w Verilogu.

```
1 module seg7(
2   input wire clk_25mhz,
3   input wire [15:0] bcd,
4
5   output reg [3:0] an_led,
6   output reg [6:0] seg_led
7 );
8
9   function [6:0] num_to_seg (
10     input [3:0] num
11   );
12     case (num)
13       4'h0: num_to_seg = 7'b1000000;
14       4'h1: num_to_seg = 7'b1111001;
15       4'h2: num_to_seg = 7'b0100100;
16       4'h3: num_to_seg = 7'b0110000;
17       4'h4: num_to_seg = 7'b0011001;
18       4'h5: num_to_seg = 7'b0010010;
19       4'h6: num_to_seg = 7'b0000010;
20       4'h7: num_to_seg = 7'b1111000;
21       4'h8: num_to_seg = 7'b0000000;
22       4'h9: num_to_seg = 7'b0010000;
23       4'ha: num_to_seg = 7'b0001000;
24       4'hb: num_to_seg = 7'b0000011;
25       4'hc: num_to_seg = 7'b1000110;
26       4'hd: num_to_seg = 7'b0100001;
27       4'he: num_to_seg = 7'b0000110;
28       4'hf: num_to_seg = 7'b0001110;
29     endcase
30   endfunction
31
32   reg [16:0] counter;
33   wire [1:0] selector;
34
35   // Switch each segment at frequency 25Mhz / 2^15 ~ 763Hz.
```

```

36     assign selector = counter[16:15];
37
38     always @ (posedge clk_25mhz) begin
39         counter <= counter + 1;
40
41         case (selector)
42             2'b00: begin
43                 an_led <= 4'b0111;
44                 seg_led <= num_to_seg(bcd[15:12]);
45             end
46             2'b01: begin
47                 an_led <= 4'b1011;
48                 seg_led <= num_to_seg(bcd[11:8]);
49             end
50             2'b10: begin
51                 an_led <= 4'b1101;
52                 seg_led <= num_to_seg(bcd[7:4]);
53             end
54             2'b11: begin
55                 seg_led <= num_to_seg(bcd[3:0]);
56                 an_led <= 4'b1110;
57             end
58         endcase
59     end
60 endmodule

```

Listing 2: Kod modułu seg7 w Verilogu.

```

1  module main (
2      input wire CLK_25MHZ,
3      input wire [7:0] SW,
4      input wire [1:0] BTN,
5
6      output wire [7:0] LED,
7      output wire [3:0] AN_LED,
8      output wire [6:0] SEG_LED
9  );
10
11     reg [11:0] bin = 0;
12     wire [15:0] bcd;
13
14     assign LED = SW;
15
16     bin2bcd bin2bcd(
17         .bin(bin),
18         .bcd(bcd)
19     );
20
21     seg7 seg7(
22         .clk_25mhz(CLK_25MHZ),
23         .an_led(AN_LED),
24         .seg_led(SEG_LED),
25         .bcd(bcd)
26     );
27
28     always @ (posedge CLK_25MHZ) begin
29         if (BTN[0]) begin
30             bin <= SW[7:4] + SW[3:0];
31         end else if (BTN[1]) begin
32             bin <= SW[7:4] * SW[3:0];
33         end else begin
34             bin <= SW[7:4] * 100 + SW[3:0];

```

```

35     end
36 end
37 endmodule

```

Listing 3: Kod głównego modułu w Verilogu.

```

1  // Basys 2 pins configuration.
2
3  NET "CLK_25MHZ" LOC = "B8";
4
5  // NET "BTN<3>" LOC = "A7";
6  // NET "BTN<2>" LOC = "M4";
7  NET "BTN<1>" LOC = "C11";
8  NET "BTN<0>" LOC = "G12";
9
10 NET "SW<7>" LOC = "N3";
11 NET "SW<6>" LOC = "E2";
12 NET "SW<5>" LOC = "F3";
13 NET "SW<4>" LOC = "G3";
14 NET "SW<3>" LOC = "B4";
15 NET "SW<2>" LOC = "K3";
16 NET "SW<1>" LOC = "L3";
17 NET "SW<0>" LOC = "P11";
18
19 NET "LED<7>" LOC = "G1" ;
20 NET "LED<6>" LOC = "P4" ;
21 NET "LED<5>" LOC = "N4" ;
22 NET "LED<4>" LOC = "N5" ;
23 NET "LED<3>" LOC = "P6" ;
24 NET "LED<2>" LOC = "P7" ;
25 NET "LED<1>" LOC = "M11";
26 NET "LED<0>" LOC = "M5" ;
27
28 NET "SEG_LED<0>" LOC = "L14";
29 NET "SEG_LED<1>" LOC = "H12";
30 NET "SEG_LED<2>" LOC = "N14";
31 NET "SEG_LED<3>" LOC = "N11";
32 NET "SEG_LED<4>" LOC = "P12";
33 NET "SEG_LED<5>" LOC = "L13";
34 NET "SEG_LED<6>" LOC = "M12";
35
36 NET "AN_LED<3>" LOC = "K14";
37 NET "AN_LED<2>" LOC = "M13";
38 NET "AN_LED<1>" LOC = "J12";
39 NET "AN_LED<0>" LOC = "F12";

```

Listing 4: Konfiguracja pinów płytki Basys 2.
